EXAMPLE 1.6-2: Thermoelectric Heat Sink

Heat rejection from a thermoelectric cooling device is accomplished using a 10x10 array of D_{fin} = 1.5 mm diameter pin fins that are L_{fin} = 15 mm long. The fins are attached to a square base plate that is W_b = 3 cm on a side and th_b = 2 mm thick, as shown in Figure 1. The conductivity of the fin material is k_{fin} = 70 W/m-K and the thermal conductivity of the base material is k_b = 25 W/m-K. There is a contact resistance of R''_c = 1x10^{-4} m^2-K/W at the interface between the base of the fins and the base plate. The hot end of the thermoelectric cooler is at T_{hot} = 30ºC and the surrounding air temperature is T_∞ = 20ºC. The average heat transfer coefficient between the air and the surface of the heat sink is $\overline{h}$ = 50 W/m^2-K.



Figure 1: Heat sink mounted on a thermoelectric cooler

a.) What is the total thermal resistance between the hot end of the thermoelectric cooler and the air? What is the rate of heat rejection that can be accomplished under these conditions?

The first section of the EES code provides the inputs for the problem.

```
"EXAMPLE 1.6-2: Thermoelectric Heat Sink"
$UnitSystem SI MASS RAD PA K J
$Tabstops 0.2 0.4 0.6 3.5 in

"Inputs"
T_infinity=converttemp(C,K,20 [C])          "Air temperature"
T_hot=converttemp(C,K,30 [C])               "Hot end of thermoelectric cooler"
D_fin=1.5[mm]*convert(mm,m)                 "Fin diameter"
L_fin=15[mm]*convert(mm,m)                  "Fin length"
N_fin=100                                   "Number of fins"
W_b=3 [cm] *convert(cm,m)                   "Width of base (square)"
th_b=2[mm]*convert(mm,m)                    "Thickness of base"
k_fin=70 [W/m-K]                            "Conductivity of fin"
k_b=25 [W/m-K]                              "Conductivity of base"
h_bar=50 [W/m^2-K]                          "Heat transfer coefficient"
R``_c=1e-4 [m^2-K/W]                        "Contact resistance"
```

The constant cross-sectional area fins can be treated using the solutions presented in Section 1.6. The perimeter (*per*), cross-sectional area (A_c), and surface area for convection ($A_{s,fin}$, assuming adiabatic ends) associated with each fin are calculated according to:

$$per = \pi\, D_{fin}$$

$$A_c = \frac{\pi}{4} D_{fin}^2$$

$$A_{s,fin} = \pi\, L\, D_{fin}$$

```
per=pi*d_fin                          "Perimeter of fin"
A_c=pi*D_fin^2/4                      "Cross-sectional area for conduction"
A_s_fin=pi*L_fin*D_fin                "Surface area of fin for convection"
```

The fin constant and fin efficiency for an adiabatic tip, constant cross-sectional area fin are computed according to:

$$m = \sqrt{\frac{per\, \bar{h}}{k_{fin}\, A_c}}$$

$$\eta_{fin} = \frac{\tanh\left(m\, L_{fin}\right)}{m\, L_{fin}}$$

The resistance of any type of fin (R_{fin}) can be obtained from its efficiency:

$$R_{fin} = \frac{1}{\eta_{fin}\, \bar{h}\, A_{s,fin}}$$

```
mL=sqrt(h_bar*per/(k_fin*A_c))*L_fin  "Fin parameter"
eta_fin=tanh(mL)/mL                   "Fin efficiency"
R_fin=1/(h_bar*A_s_fin*eta_fin)       "Fin resistance"
```

The resistance network that represents the entire heat sink (Figure 2) extends from the hot end of the cooler to the air and includes conduction through the base ($R_{cond,b}$) followed by two paths in parallel corresponding to the heat that is transferred by convection from the unfinned upper surface of the base ($R_{un\text{-}finned}$) and the heat that is transferred through the contact resistance at the base of the fins (R_c) and then through the resistance associated with the fin itself (R_{fin}). Note that R_c and R_{fin} are in parallel N_{fin} times and therefore the value these resistances in the circuit is reduced by $1/N_{fin}$.



Figure 2: Resistance network representing the heat sink

The resistance to conduction through the base is:

$$R_{cond,b} = \frac{th_b}{k_b\, W_b^2}$$

The contact resistance associated with each fin-to-base interface is:

$$R_c = \frac{R''_c}{A_c}$$

The resistance of the unfinned region of the base is:

$$R_{un-finned} = \frac{1}{\bar{h}\left(W_b^2 - N_{fin}\, A_c\right)}$$

These resistances are calculated in EES:

```
R_b=th_b/(k_b*W_b^2)                    "Resistance due to conduction through the base"
R_unfinned=1/((W_b^2-N_fin*A_c)*h_bar)  "Resistance of unfinned base"
R_c=R``_c/A_c                           "Fin-to-base contact resistance"
```

The total resistance (R_{tot}) and heat transfer ($\dot{q}_{tot}$) from the heat sink are obtained according to:

$$R_{tot} = R_b + \left(\frac{1}{\left(\frac{R_c}{N_{fin}} + \frac{R_{fin}}{N_{fin}}\right)} + \frac{1}{R_{un-finned}} \right)^{-1}$$

$$\dot{q}_{tot} = \frac{\left(T_{hot} - T_{\infty}\right)}{R_{tot}}$$

and calculated in EES.

```
R_tot=R_b+(1/(R_c/N_fin+R_fin/N_fin)+1/R_unfinned)^(-1)          "Total resistance"
q_dot_tot=(T_hot-T_infinity)/R_tot                                "Total rate of heat transfer"
```

The result is a total resistance of 3.42 K/W and the rate of heat transfer is 2.92 W. The numerical values of each resistance are included in Figure 2 in order to understand the mechanisms that are governing the behavior of the heat sink. Notice that the resistance of the base is not very important, as it is a small resistor in series with larger ones. The resistance of the unfinned portion of the base is also not critical, since it is a large resistor in parallel with smaller ones. On the other hand, both the contact resistance and the fin resistance are important as these two resistors dominate the problem and are of the same order of magnitude. The fin resistance is the most critical parameter in the problem and any attempt to improve performance should focus on this element of the heat sink.

b.) Through material selection and manipulation of the air flow across the heat sink, it is possible to affect design changes to k_{fin} and $\bar{h}$. Generate a contour plot that illustrates contours of constant heat rejection in the parameter space of k_{fin} (ranging from 5 W/m-K to 150 W/m-K) and $\bar{h}$ (ranging from 10 W/m^2-K to 200 W/m^2-K).

One of the nice things about solving problems using a computer program as opposed to pencil and paper is that parametric studies and optimization are relatively straightforward once the problem is solved. In order to prepare a contour plot with EES, it is necessary to setup a parametric table in which both of the parameters of interest vary over a specified range. Open a new parametric table and include the 2 independent variables (the variables k_fin and h_bar) as well as the dependent variable of interest (the variable q_dot_tot). In order to run the simulation for 20 values of k_{fin} and 20 values of $\bar{h}$ it is necessary to include 20x20=400 runs in the table. (Add runs using the Insert/Delete Runs option from the Tables menu.)

It is necessary to set the values of k_fin and h_bar in the table. It is possible to vary k_fin from 5 to 150 W/m-K, 20 times by using the "Repeat pattern every" option in the Alter Values dialog that appears when you right-click on the k_fin column, as shown in Figure 3.



Figure 3: Vary k_{fin} from 5 to 150 W/m-K 20 times

In order to completely cover the parameter space, it is necessary to evaluate a different value of h_bar at each unique value of k_fin; this can be accomplished using the "Apply pattern every" option in the Alter Values dialog for the h_bar column of the table, see Figure 4.



Figure 4: Vary $\bar{h}$ from 10 to 200 W/m-K with 20 runs for each of 20 values

When the specified values of the variables k_fin and h_bar are commented out in the Equations window, it is possible to run the parametric table using the Solve Table command in the Calculate menu (F3); 400 values of $\dot{q}$ are determined, one for each combination of k_{fin} and $\bar{h}$ set in the parametric table. To generate a contour plot, select X-Y-Z plot from the New Plot Window option in the Plots menu. Select k_fin as the variable on the x-axis, h_bar as the y-axis variable and q_dot_tot as the contour variable. It is possible to adjust the appearance of the resulting contour plot by altering the resolution, smoothing, color options, and the type of function used for interpolation. The contour plot using isometric lines is shown in Figure 5.



Figure 5: Contours of constant heat transfer rate in the parameter space of fin material conductivity and heat transfer coefficient.

The nominal design point shown in Figure 1 is also indicated in Figure 5. Contour plots are useful in that they can clarify the impact of design changes. For example, Figure 5 shows that it would be more beneficial to explore methods to increase the heat transfer coefficient than the fin conductivity at the nominal design conditions (i.e., moving from the nominal design point towards higher heat transfer will result in much larger performance gains than towards higher fin conductivity).